Newsgroup: it.comp.appl.access
Titolo: Unire due file access

E' possibile unire due file access in un unico data base?
Grazie infinite.
 - Bruno -

  - Bruno -
Cosa intendi....????????
Alessandro/@Alex
-- 
Posted via Mailgate.ORG Server - 
http://www.Mailgate.ORG


Certo
Vedi Importazioni e/o Collegamenti
--
Maurizio75


Vorrei riformulare la domanda. E' possibile "lanciare" l'esecuzione di una
Form posta in un altro file Accesss
diverso da quello in esecuzione?
L'obiettivo è quello di "lavorare :-))" con due distinti data base senza
dover uscire e ricaricare.
Scusate il liguaggio (non proprio tecnico). Grazie
- Bruno -

 - Bruno -
Doverosa precisazione....!
Inserisci il codice in un Modulo Generico, e richiama la
Funzione fOpenRemoteForm passando la Path completa del DB
e il nome della Form da aprire....!!!
Ciao.
Alessandro/@Alex
Private Declare Function apiSetForegroundWindow Lib "user32" _
            Alias "SetForegroundWindow" _
            (ByVal hwnd As Long) _
            As Long
Private Declare Function apiShowWindow Lib "user32" _
            Alias "ShowWindow" _
            (ByVal hwnd As Long, _
            ByVal nCmdShow As Long) _
            As Long
Private Const SW_MAXIMIZE = 3
Private Const SW_NORMAL = 1
Function fOpenRemoteForm(strMDB As String, _
                                        strForm As String, _
                                        Optional intView As Variant) _
                                        As Boolean
Dim objAccess As Access.Application
Dim lngRet As Long
    On Error GoTo fOpenRemoteForm_Err
    If IsMissing(intView) Then intView = acViewNormal
    If Len(Dir(strMDB)) > 0 Then
        Set objAccess = New Access.Application
        With objAccess
            lngRet = apiSetForegroundWindow(.hWndAccessApp)
            lngRet = apiShowWindow(.hWndAccessApp, SW_NORMAL)
            'the first call to ShowWindow doesn't seem to do anything
            lngRet = apiShowWindow(.hWndAccessApp, SW_NORMAL)
            .OpenCurrentDatabase strMDB
            .DoCmd.OpenForm strForm, intView
            Do While Len(.CurrentDb.Name) > 0
                DoEvents
            Loop
        End With
    End If
fOpenRemoteForm_Exit:
    On Error Resume Next
    objAccess.Quit
    Set objAccess = Nothing
    Exit Function
fOpenRemoteForm_Err:
    fOpenRemoteForm = False
    Select Case Err.Number
        Case 7866:
            'mdb is already exclusively opened
            MsgBox "The database you specified " & vbCrLf & strMDB & _
                vbCrLf & "is currently open in exclusive mode.  " &
vbCrLf _
                & vbCrLf & "Please reopen in shared mode and try again",
_
                vbExclamation + vbOKOnly, "Could not open database."
        Case 2102:
            'form doesn't exist
            MsgBox "The Form '" & strForm & _
                        "' doesn't exist in the Database " _
                        & vbCrLf & strMDB, _
                        vbExclamation + vbOKOnly, "Form not found"
        Case 7952:
            'user closed mdb
            fOpenRemoteForm = True
        Case Else:
            MsgBox "Error#: " & Err.Number & vbCrLf & Err.Description, _
                    vbCritical + vbOKOnly, "Runtime error"
    End Select
    Resume fOpenRemoteForm_Exit
End Function
'************ Code End *************
